NAG Fortran Library Routine Document

S18GKF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

S18GKF returns a sequence of values for the Bessel functions $J_{\alpha+n-1}(z)$ or $J_{\alpha-n+1}(z)$ for complex z, non-negative $\alpha < 1$ and $n = 1, 2, \ldots, |N| + 1$.

2 Specification

```
SUBROUTINE S18GKF (Z, A, NL, B, IFAIL)
INTEGER              NL, IFAIL
double precision     A
complex*16           Z, B(*)
```

3 Description

S18GKF evaluates a sequence of values for the Bessel function of the first kind $J_\alpha(z)$, where z is complex and non-zero and α is the order with $0 \leq \alpha < 1$. The $(|N| + 1)$-member sequence is generated for orders $\alpha, \alpha + 1, \ldots, \alpha + |N|$ when $N \geq 0$. Note that $+$ is replaced by $-$ when $N < 0$. For positive orders the routine may also be called with $z = 0$, since $J_q(0) = 0$ when $q > 0$. For negative orders the formula

$$J_{-q}(z) = \cos(\pi q) J_q(z) - \sin(\pi q) Y_q(z)$$

is used to generate the required sequence. The appropriate values of $J_q(z)$ and $Y_q(z)$ are obtained by calls to S17DEF and S17DCF.

4 References

Abramowitz M and Stegun I A (1972) *Handbook of Mathematical Functions* (3rd Edition) Dover Publications

5 Parameters

1: Z – ***complex*16*** *Input*

On entry: the argument z of the function.

Constraint: $\mathrm{Z} \neq (0.0, 0.0)$ when $\mathrm{NL} < 0$.

2: A – ***double precision*** *Input*

On entry: the order α of the first member in the required sequence of function values.

Constraint: $0.0 \leq \mathrm{A} < 1.0$.

3: NL – INTEGER *Input*

On entry: the value of N.

Constraint: $\mathrm{abs(NL)} \leq 101$.

4: B(∗) – ***complex*16*** array *Output*

Note: the dimension of the array B must be at least abs(NL) + 1.

On exit: with IFAIL = 0 or 3, the required sequence of function values: B(n) contains $J_{\alpha+n-1}(z)$ if NL ≥ 0 and $J_{\alpha-n+1}(z)$ otherwise, for $n = 1, 2, \ldots,$ abs(NL) + 1.

5: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value −1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value −1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, Z = (0.0, 0.0) when NL < 0,
or A < 0.0,
or A ≥ 1.0,
or abs(NL) > 101.

IFAIL = 2

The computation has been abandoned due to the likelihood of overflow.

IFAIL = 3

The computation has been completed but some precision has been lost.

IFAIL = 4

The computation has been abandoned due to total loss of precision.

IFAIL = 5

The computation has been abandoned due to failure to satisfy the termination condition.

7 Accuracy

All constants in S17DCF and S17DEF are specified to approximately 18 digits of precision. If t denotes the number of digits of precision in the floating-point arithmetic being used, then clearly the maximum number of correct digits in the results obtained is limited by $p = \min(t, 18)$. Because of errors in argument reduction when computing elementary functions inside S17DCF and S17DEF, the actual number of correct digits is limited, in general, by $p - s$, where $s \approx \max(1, |\log_{10}|z||, |\log_{10}|\alpha||)$ represents the number of digits lost due to the argument reduction. Thus the larger the values of $|z|$ and $|\alpha|$, the less the precision in the result.

8 Further Comments

None.

9 Example

The example program evaluates $J_0(z), J_1(z), J_2(z)$ and $J_3(z)$ at $z = 0.6 - 0.8i$, and prints the results.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     S18GKF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NLMAX
      PARAMETER        (NLMAX=102)
      DOUBLE PRECISION ONE
      PARAMETER        (ONE=1.0D+0)
*     .. Local Scalars ..
      COMPLEX *16      Z
      DOUBLE PRECISION A, ALPHA
      INTEGER          I, IFAIL, NL
*     .. Local Arrays ..
      COMPLEX *16      B(NLMAX)
*     .. External Subroutines ..
      EXTERNAL         S18GKF
*     .. Intrinsic Functions ..
      INTRINSIC        ABS, DBLE, SIGN
*     .. Executable Statements ..
      WRITE (NOUT,*) 'S18GKF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
   20 READ (NIN,*,END=60) Z, A, NL
      IFAIL = 0
*
      CALL S18GKF(Z,A,NL,B,IFAIL)
*
      WRITE (NOUT,*)
      WRITE (NOUT,*) '      Z           A     NL    IFAIL'
      WRITE (NOUT,*)
      WRITE (NOUT,99999) Z, A, NL, IFAIL
      WRITE (NOUT,*)
      WRITE (NOUT,*) ' Requested values of J_alpha(Z)'
      WRITE (NOUT,*)
      ALPHA = A
      WRITE (NOUT,*) '     alpha              J_alpha(Z)'
      DO 40 I = 1, ABS(NL) + 1
         WRITE (NOUT,99998) ALPHA, B(I)
         ALPHA = ALPHA + SIGN(ONE,DBLE(NL))
   40 CONTINUE
      GO TO 20
   60 STOP
*
99999 FORMAT (1X,'( ',F4.1,', ',F4.1,' )',2X,F4.1,I6,I7)
99998 FORMAT (1X,1P,D12.4,3X,'(',D12.4,',',D12.4,' )')
      END
```

9.2 Program Data

```
S18GKF Example Program Data
 ( 0.6,-0.8)   0.0   3 : Values of Z, A and NL
```

9.3 Program Results

```
 S18GKF Example Program Results

       Z           A     NL    IFAIL

(  0.6, -0.8 )   0.0      3       0

  Requested values of J_alpha(Z)

     alpha              J_alpha(Z)
  0.0000D+00   (  1.0565D+00,  2.4811D-01 )
  1.0000D+00   (  3.5825D-01, -3.7539D-01 )
  2.0000D+00   ( -2.5974D-02, -1.2538D-01 )
  3.0000D+00   ( -1.9369D-02, -8.6380D-03 )
```